

Introduction à METAPOST

Victor SANNIER
Association GUTenberg
Décembre 2023

Les exposés mensuels
GUTenberg

Plan de la présentation

Les exposés mensuels
GUTenberg

I. Introduction

II. Concepts de base

III. Programmation impérative

IV. Programmation déclarative

V. Paquets

VI. Ressources utiles

METAPOST en deux mots

Les exposés mensuels
GUTenberg

METAPOST est un

- langage pour le dessin vectoriel (comme TikZ, Asymptote, etc.),
- construit autour d'un solveur d'équations linéaires,
- populaire dans les communautés Plain T_EX et ConT_EXt.

Historique du projet

Les exposés mensuels
GUTenberg

- 1979. Donald KNUTH publie une première version de METAFONT et de T_EX.
- 1984. Donald KNUTH publie une révision complète des deux programmes.
- 1994. John D. HOBBY développe METAPOST.
- 1990. Hans HAGEN travaille sur ConT_EXt, qui inclut une extension de METAPOST.

Utilisation de METAPOST

Les exposés mensuels
GUTenberg

Avec T_EX (à la Knuth), après avoir exécuté `mpost circle.mp`:

```
\input epsf
\epsfbox{circle.1}
\bye
```

Avec ConT_EXt:

```
\starttext
  \startMPcode
    draw fullcircle scaled 1cm ;
  \stopMPcode
\stoptext
```

Avec LuaL^AT_EX:

```
\usepackage{luamplib}
\begin{document}
  \begin{mplibcode}
    draw fullcircle scaled 1cm ;
  \end{mplibcode}
\end{document}
```

Dans tous les cas, on obtient: 

Remarque. METAPOST peut aussi créer des fichiers EPS, PDF, SVG, PNG, etc. utilisables par d'autres logiciels.

```
outputformat := "svg";
outputtemplate := "%j-%c.svg";
```

Les types de données

Les exposés mensuels
GUTenberg

- numeric et boolean,
- pair et color,
- path et picture,
- pen,
- string,
- transform (translation, rotation, homothétie, ...).

```
path p;  
p := (0, 0) -- (1, 1) -- (2, 0) -- cycle;
```

Les courbes de BÉZIER


Les exposés mensuels
GUTenberg

MetaPost interpole entre les points de contrôle pour dessiner une courbe lisse.

```
z0 = (0, 0); z1 = (60, 40); z2 = (40, 90);  
z3 = (10, 70); z4 = (30, 50);
```

```
label.bot(btex $z_0$ etex, z0);  
label.bot(btex $z_4$ etex, z4);
```

```
draw z0 .. z1 .. z2 .. z3 .. z4;  
draw (z0 .. z1{up} .. z2{left} .. z3 .. z4)  
  shifted (90, 0);
```



Deux paradigmes de programmation


Les exposés mensuels
GUTenberg

Programmation impérative. Définition puis transformation d'objets
(opérateur :=, boucles for).

Programmation déclarative. Expression de relations entre les objets
(opérateur =).

Exemple: Hexagone régulier


```
r := 4cm;  
  
draw fullcircle scaled (2r);  
% draw fullcircle scaled (sqrt(3)*r);  
  
draw (r, 0)  
for i=1 step 1 until 5:  
  -- r * (cosd(60i), sind(60i))  
endfor  
-- cycle;
```



Exemple: Drapeau tricolore

```
numeric w, h;  
h := 6cm;  
w := h * 3/2;  
  
fill unitsquare xyscaled (w/3, h)  
  withcolor \MPcolor{frenchblue};  
  
fill unitsquare xyscaled (w/3, h)  
  shifted (2w/3, 0)  
  withcolor \MPcolor{frenchred};
```

Les exposés mensuels
GUTenberg



Le rôle des équations linéaires

Parfois, nous ne souhaitons pas donner les coordonnées de tous les points de contrôle. À la place, nous pouvons donner des équations linéaires qui décrivent les relations entre les objets de notre dessin.

```
numeric a, b;  
a + b = 3; 2a = b + 3;  
show a, b;
```


Le solveur peut travailler avec des nombres, des points et des transformations.

Exemple: Milieu d'un segment

```
z0 = (0, 0); z1 = (6cm, 2cm);  
draw z0 -- z1;  
dotlabel.lft(btex $z_0$ etex, z0);  
dotlabel.rt(btex $z_1$ etex, z1);
```


```
z2 = (z0 + z1) / 2;  
dotlabel.bot(btex $z_2$ etex, z2);
```

Les exposés mensuels
GUTenberg




Exemple: Triangle équilatéral

```
z0 = (4cm, 0);  
z1 = z0 rotated 120;  
z2 = z0 rotated -120;  
  
fill z0 .. z1 .. z2 .. cycle  
  withcolor darkred;  
  
fill z0 -- z1 -- z2 -- cycle  
  withcolor white;
```



Exemple: Fractale de carrés

```
s := 8cm; d := 1cm;  
transform t; path p;  
(0, 0) transformed t = (d, 0);  
(0, s) transformed t = (0, s - d);  
(s, 0) transformed t = (s, d);  
p := (0, 0) -- (s, 0) -- (s, s) -- (0, s)  
    -- cycle;  
  
for i := 0 upto 12:  
    draw p; p := p transformed t;  
endfor;
```



Procédures (ou *macros*)

Les exposés mensuels
GUTenberg

expr est le type d'une expression METAPOST quelconque.


```
vardef draw_square(expr x, y, s) =  
  draw (x, y) -- (x + s, y) --  
    (x + s, y + s) -- (x, y + s) -- cycle;  
enddef;
```

Pour créer une variable locale, on utilise save.

```
def whatever = begingroup save x; x endgroup;
```


Exemple: Fractale de Sierpiński

```
vardef sierpinski(expr A, B, C, n) =  
  if n > 0:  
    sierpinski(A, 1/2[A,B], 1/2[A,C], n-1);  
    sierpinski(1/2[A,B], B, 1/2[B,C], n-1);  
    sierpinski(1/2[A,C], 1/2[B,C], C, n-1);  
  else:  
    fill A--B--C--cycle withcolor black;  
  fi;  
enddef;  
  
sierpinski((0, 0), (0, 8cm), (8cm, 8cm), 8);
```



De nombreux paquets

Les exposés mensuels
GUTenberg

Voir notamment <https://ctan.org/topic/graphics-mpost>


- automata par Gabriele PUPPIS,
- boxes par John D. HOBBY,
- byrne et fiziko par Sergey SLYUSAREV,
- drv par Laurent MÉHATS,
- huffman et MPchess par Maxime CHUPIN,
- metago par Wentao ZHENG,
- mptrees et repere par Olivier PÉAULT,
- etc.

Le paquet fiziko

```
input fiziko.mp;
numeric w, h, n;
w := 8cm; h := 6cm; n := 5;
radius = w / 2n;
springwidth := radius / 3;

draw woodBlock(w, h / 4);
for i = 1 upto n:
  x := radius * (2i - 1);
  draw sphere.c(2 * radius) shifted (x, -h);
  draw spring((x, radius - h), (x, 0), 5 + i);
endfor;
```


Les exposés mensuels
GUTenberg



Le paquet metago

```
input metago.mp;  
  
set_board_size 9;  
init_board;  
  
put_b(C4); put_b(D3); put_b(E4);  
put_w(D4); put_w(C3); put_w(B4);  
put_b(I9); put_w(I8);  
  
display_board;
```

Les exposés mensuels
GUTenberg



Manuels et Sites web

Les exposés mensuels
GUTenberg

- *METAPOST: A User's Manual* par John D. HOBBY
Disponible en traduction française.
- *Learning METAPOST by Doing* par André HECK,
- *Exemples METAPOST* par l'association GUTenberg
<https://metapost.gutenberg-asso.fr/>
- De nombreux articles publiés dans la revue TUGBoat et dans la Lettre GUTenberg.